Logic Languages

Paul Fodor
CSE260, Computer Science B: Honors
Stony Brook University

http://www.cs.stonybrook.edu/~cse260
Languages

• Paradigms of Programming Languages:
  • Imperative = Turing machines
  • Functional Programming = lambda calculus
  • Logical Programming = first-order predicate calculus

• Prolog and its variants make up the most commonly used Logical programming languages.
  • One variant is XSB Prolog (developed here at Stony Brook)
  • Other Prolog systems: SWI Prolog, Sicstus, Yap Prolog, Ciao Prolog, GNU Prolog, etc.
    • ISO Prolog standard.
Relations/Predicates

- Predicates are building-blocks in predicate calculus: \( p(a_1, a_2, \ldots, a_k) \)

- \( \text{parent}(X, Y) \): \( X \) is a parent of \( Y \).
  
  \[
  \begin{align*}
  \text{parent}(\text{pam}, \text{bob}). & \quad \text{parent}(\text{bob}, \text{ann}). \\
  \text{parent}(\text{tom}, \text{bob}). & \quad \text{parent}(\text{bob}, \text{pat}). \\
  \text{parent}(\text{tom}, \text{liz}). & \quad \text{parent}(\text{pat}, \text{jim}).
  \end{align*}
  \]

- \( \text{male}(X) \): \( X \) is a male.
  
  \[
  \begin{align*}
  \text{male}(\text{tom}). \\
  \text{male}(\text{bob}). \\
  \text{male}(\text{jim}).
  \end{align*}
  \]

We attach meaning to them, but within the logical system they are simply structural building blocks, with no meaning beyond that provided by explicitly-stated interrelationships.
Relations

- **female(X)**: X is a female.

  female(pam).
  female(pat).
  female(ann).
  female(liz).
parent(pam, bob).
parent(tom, bob).
parent(tom, liz).
parent(bob, ann).
parent(bob, pat).
parent(pat, jim).
female(pam).
female(pat).
female(ann).
female(liz).
male(tom).
male(bob).
male(jim).
Relations

- Rules:
  - \texttt{mother}(X, Y): X is the mother of Y.

    - In First Order Logic (FOL or predicate calculus):
      \[ \forall X, Y \ (\text{parent}(X, Y) \land \text{female}(X) \implies \text{mother}(X, Y)) \]

    - In Prolog:
      \begin{verbatim}
      mother(X, Y) :-
        parent(X, Y),
        female(X).
      \end{verbatim}

  - all variables are universally quantified outside the rule
  - \texttt{“,”} means \textit{and} (conjunction), \texttt{“:-”} means \textit{if} (implication) and \texttt{“;”} means \textit{or} (disjunction).
Relations

- More Relations:

\[
\text{grandparent}(X, Y) : - \\
\text{parent}(X, Z), \\
\text{parent}(Z, Y).
\]
Relations

\[
sibling(X,Y) :- \text{parent}(Z,X), \text{parent}(Z,Y), X \neq Y.
\]

?- sibling(ann,Y).
Relations

More Relations:

cousin(X,Y) :- ...
greatgrandparent(X,Y) :- ...
greatgreatgrandparent(X,Y) :- ...

Recursion

ancestor(X,Y) :-
    parent(X,Y).

ancestor(X,Y) :-
    parent(X,Z),
    ancestor(Z,Y).

?- ancestor(X,jim).
?- ancestor(pam,X).
?- ancestor(X,Y).
Relations

• How to implement “I'm My Own Grandpa”?

https://www.youtube.com/watch?v=eYlJH81dSiw
Recursion

• What about:

\[
\text{ancestor}(X,Y) : - \\
\text{ancestor}(X,Z), \\
\text{parent}(Z,Y).
\]

\[
\text{ancestor}(X,Y) : - \\
\text{parent}(X,Y).
\]

?- \text{ancestor}(X,Y).

\text{INFINITE LOOP}
mother(X,Y):-
    parent(X,Y), female(X).

?- mother(M, bob).
?- parent(M, bob), female(M).
?- M=pam, female(pam).

M = pam        true

?- father(M, bob).
?- parent(M, bob), male(M)
   (i) ?- M=pam, male(pam).
      fail
   (ii) ?- M=tom, male(tom).

M = tom        true
Prolog Execution

• Call: Call a predicate (invocation)
• Exit: Return an answer to the caller
• Fail: Return to caller with no answer
• Redo: Try next path to find an answer
The XSB Prolog System

- [http://xsb.sourceforge.net](http://xsb.sourceforge.net)
  - Developed at Stony Brook by David Warren and many contributors

- Overview of Installation:
  - Unzip/untar; this will create a subdirectory XSB
  - Windows: you are done
  - Linux:
    ```
    cd XSB/build
    ./configure
    ./makexsb
    That's it!
    ```
  - Cygwin under Windows: same as in Linux
Use of XSB

- Put your ruleset and data in a file with extension .P (or .pl)
  
  \[
  p(X) :- q(X,\_).
  q(1,a) .
  q(2,a) .
  q(b,c) .
  \]

- Don’t forget: all rules and facts end with a period (.)

- Comments: /*...*/ or %.... (% acts like // in Java/C++)

- Type
  
  \[
  \ldots/XSB/bin/xsb \quad (Linux/Cygwin)
  \ldots\XSB\config\x86-pc-windows\bin\xsb \quad (Windows)
  \]
  
  where \ldots is the path to the directory where you downloaded XSB

- You will see a prompt

  \[
  | ?- 
  \]

  and are now ready to type queries
Use of XSB

- Loading your program, myprog.P or myprog.pl

```prolog
?- [myprog].
```

XSB will compile myprog.P (if necessary) and load it.

Now you can type further queries, e.g.

```prolog
?- p(X).
?- p(1).
```

- Some Useful Built-ins:
  - `write(X)` – write whatever X is bound to
  - `writeln(X)` – write then put newline
  - `nl` – output newline
  - Equality: `=`
  - Inequality: `\=`

http://xsb.sourceforge.net/manual1/index.html (Volume 1)
Use of XSB

• Some Useful Tricks:
  • XSB returns only the first answer to the query
  • To get the next, type ; <Return>. For instance:
    
    | ?- q(X).
    X = 2;
    X = 4
    yes

  • Usually, typing the ;’s is tedious. To do this programmatically, use this idiom:

    | ?- (q(_X), write('X='), writeln(_X), fail ; true).

_X here tells XSB to not print its own answers, since we are printing them by ourselves. (XSB won’t print answers for variables that are prefixed with a _.)
In logic, most statements can be written many ways.

That's great for people but a nuisance for computers.

It turns out that if you make certain restrictions on the format of statements you can prove theorems mechanically.

Most common restriction is to have a single conclusion implied by a conjunction of premises (i.e., Horn clauses)

Horn clauses are named for the logician Alfred Horn, who first pointed out their significance in 1951.

That's what logic programming systems do!
Syntax of Prolog Programs

• A Prolog program is a sequence of clauses

• Each clause (sometimes called a rule or Horn rule) is of the form:

   Head :- Body.

   • Head is one term
   • Body is a comma-separated list of terms
   • A clause with an empty body is called a fact
Logic Programming Concepts

- Operators:
  - conjunction, disjunction, negation, implication
- Universal and existential quantifiers
- Statements
  - sometimes true, sometimes false, sometimes unknown
  - axioms - assumed true
  - theorems - provably true
  - goals - things we'd like to prove true
• A **term** can be a **constant**, **variable**, or **structure** (consisting of a **functor** and a parenthesized list of arguments)

• A **constant** is either an **atom** or a **number**
  • An **atom** is either what looks like an identifier **beginning with** a lowercase letter, or a **single quoted string**
  • A **number** looks like an integer or real from some more ordinary language

• A **variable** looks like an identifier **beginning with** an upper-case letter

• There are NO declarations (vars, terms, or predicates)
  • All types are discovered implicitly
Logic Programming Concepts

- The Prolog interpreter has a collection of facts and rules in its DATABASE
  - Facts (i.e., clauses with empty bodies):
    - `raining(ny).`  `raining(seattle).`
  - Facts are axioms (things the interpreter assumes to be true)
    - Prolog provides an automatic way to deduce true results from facts and rules
  - A rule (i.e., a clause with both sides):
    - `wet(X) :- raining(X).`
  - The meaning of a rule is that the conjunction of the structures in the body implies the head.
  - Note: Single-assignment variables: X must have the same value on both sides.
  - Query or goal (i.e., a clause with an empty head):
    - `?- wet(X).`
So, rules are theorems that allow the interpreter to infer things.

To be interesting, rules generally contain variables:

\[
\text{employed}(X) \ :- \ \text{employs}(Y,X).
\]

This can be read as:

"for all X, X is employed if there exists a Y such that Y employs X""

- Note the direction of the implication.
- Also, the example does NOT say that X is employed ONLY IF there is a Y that employs X.

- there can be other ways for people to be employed:
  - like, we know that someone is employed, but we don't know who is the employer or maybe they are self employed:

\[
\text{employed}(bill).
\]
The scope of a variable is the clause in which it appears:

- Variables whose first appearance is on the left hand side of the clause (i.e., the head) have implicit universal quantifiers:
  - For example, we infer for all possible $X$ that they are employed:
    \[ \text{employed}(X) \leftarrow \text{employs}(Y,X). \]

- Variables whose first appearance is in the body of the clause have implicit existential quantifiers in that body:
  - For example, there exists some $Y$ that employs $X$
  - Note that these variables are also universally quantified outside the rule (by logical equivalences)
grandmother(A, C) :-
    mother(A, B),
    mother(B, C).

can be read as:
"for all A, C [A is the grandmother of C if there exists a B such that A is the mother of B and B is the mother of C]"

• We probably want another rule that says:

grandmother(A, C) :-
    mother(A, B),
    father(B, C).
Recursion

- Transitive closure:

Example: a graph declared with facts (true statements)

\[
\begin{align*}
\text{edge} & (1, 2). \\
\text{edge} & (2, 3). \\
\text{edge} & (2, 4).
\end{align*}
\]

1) if there's an \texttt{edge} from \texttt{X} to \texttt{Y}, we can \texttt{reach Y} from \texttt{X}:

\[
\text{reach}(X,Y) :- \text{edge}(X,Y).
\]

2) if there's an \texttt{edge} from \texttt{X} to \texttt{Z}, and we can \texttt{reach Y} from \texttt{Z}, then we can \texttt{reach Y} from \texttt{X}:

\[
\text{reach}(X,Y) :- \\
\text{edge}(X,Z), \\
\text{reach}(Z, Y).
\]
?- reach(X,Y).

\[
\begin{align*}
X &= 1 \\
Y &= 2 ; \\
X &= 2 \\
Y &= 3 ; \\
X &= 2 \\
Y &= 4 ; \\
X &= 1 \\
Y &= 3 ; \\
X &= 1 \\
Y &= 4 ; \\
\text{no}
\end{align*}
\]

\[\text{Type a semi-colon repeatedly for more answers}\]

\[
\text{reach}(X,Y) \leftarrow \text{edge}(X,Y).
\]

\[
\text{reach}(X,Y) \leftarrow \\
\text{edge}(X,Z), \\
\text{reach}(Z, Y).
\]
We will now explore Prolog programs in more detail:

Syntax of Prolog Programs

Terms can be:

- Atomic data
- Variables
- Structures
Atomic Data

- **Numeric constants**: integers, floating point numbers (e.g. $1024$, $-42$, $3.1415$, $6.023e23$, ...)

- **Atoms**:
  - Identifiers: sequence of letters, digits, underscore, beginning with a lower case letter (e.g. `paul`, `r2d2`, `one_element`).
  - Strings of characters enclosed in single quotes (e.g. `'Stony Brook'`)
Variables

- Variables are denoted by identifiers **beginning with an Uppercase letter or underscore** (e.g. X, Index, _param).

- **These are Single-Assignment Logical variables:**
  - Variables can be assigned only once
  - Different occurrences of the same variable in a clause denote the same data

- Variables are implicitly declared upon first use
  - Variables are not typed
    - All types are discovered implicitly (no declarations in LP)
  - If the variable does not start with underscore, it is assumed that it appears multiple times in the rule.
    - If it does not appear multiple times, then a warning is produced: "Singleton variable"
    - You can use variables preceded with underscore to eliminate this warning
Variables

- *Anonymous variables* (also called *Don’t care variables*): variables beginning with "_

  - Underscore, by itself (i.e., _), represents a variable
    - Each occurrence of _ corresponds to a different variable; even within a clause, _ does not stand for one and the same object.
  - A variable with a name beginning with "_", but has more characters. E.g.: _radius, _Size
    - we want to give it a descriptive name
    - sometimes it is used to create relationships within a clause (and must therefore be used more than once): a warning is produced: "Singleton-marked variable appears more than once"
Variables

• Warnings are used to identify bugs (most because of copy-paste errors)
  • Instead of declarations and type checking
  • Fix all the warnings in a program, so you know that you don't miss any logical error
Variables

- Variables can be assigned only once, but that value can be further refined:

\[ \begin{align*}
? &- \quad X = f(Y), \\
&\quad \quad Y = g(Z), \\
&\quad \quad Z = 2.
\end{align*} \]

Therefore, \[\begin{align*}
X &= f(g(2)), \quad Y = g(2), \quad Z = 2
\end{align*} \]

- The order also does not matter:

\[ \begin{align*}
? &- \quad Z = 2, \\
&\quad \quad X = f(Y), \\
&\quad \quad Y = g(Z).
\end{align*} \]

\[\begin{align*}
X &= f(g(2)), \quad Y = g(2), \quad Z = 2
\end{align*} \]

- Even infinite structures:

\[ \begin{align*}
? &- \quad X = f(X).
\end{align*} \]

\[X = f(f(f(f(f(f(f(f(f(f(f(f(f(f(f(f(f(f(\ldots)))))))))))))))))\]
To run a Prolog program, one asks the interpreter a question. This is done by asking a query which the interpreter tries to prove:

- If it can, it says *yes*
- If it can't, it says *no*
- If your query contained variables, the interpreter prints the values it had to give them to make the query true

?- wet(ny).  ?- reach(a, d).  ?- reach(d, a).
Yes Yes No
?- wet(X).  ?- reach(X, d).  ?- reach(X, Y).
X = ny;  X=a  X=a, Y=d
X = seattle; ?- reach(a, X).
no  X=d
Meaning of Logic Programs

- **Declarative Meaning:** What are the *logical consequences* of a program?
- **Procedural Meaning:** For what values of the variables in the query can I *prove* the query?

- The user gives the system a goal:
  - The system attempts to find axioms + inference rules to *prove* that goal
  - If goal contains variables, then also gives the values for those variables for which the goal is proven
Declarative Meaning

brown(bear).  big(bear).
gray(elephant).  big(elephant).
black(cat).  small(cat).
dark(Z) :- black(Z).
dark(Z) :- brown(Z).
dangerous(X) :- dark(X), big(X).

• The logical consequences of a program L is the smallest set such that
  • All facts of the program are in L,
  • If \( H :- B_1, B_2, \ldots, B_n \) is an instance of a clause in the program such that \( B_1, B_2, \ldots, B_n \) are all in L, then H is also in L.
  • For the above program we get \text{dark(cat)} \text{ and dark(bear)} \text{ and consequently dangerous(bear)} in addition to the original facts.
Procedural Meaning of Prolog

brown(bear). big(bear).
gray(elephant). big(elephant).
black(cat). small(cat).
dark(Z) :- black(Z).
dark(Z) :- brown(Z).
dangerous(X) :- dark(X), big(X).

- A query is, in general, a conjunction of goals: \( G_1, G_2, \ldots, G_n \)
- To prove \( G_1, G_2, \ldots, G_n \):
  - Find a clause \( H : - B_1, B_2, \ldots, B_k \) such that \( G_1 \) and \( H \) match.
  - Under the substitution for variables, prove \( B_1, B_2, \ldots, B_k, G_2, \ldots, G_n \)

If nothing is left to prove then the proof succeeds!
If there are no more clauses to match, the proof fails!
Procedural Meaning of Prolog

```prolog
brown(bear). big(bear).
gray(elephant). big(elephant).
black(cat). small(cat).
dark(Z) :- black(Z).
dark(Z) :- brown(Z).
dangerous(X) :- dark(X), big(X).
```

To prove: `?- dangerous(Q).`

1. Select `dangerous(X) :- dark(X), big(X)` and prove `dark(Q), big(Q).`
2. To prove `dark(Q)` select the first clause of dark, i.e.
   `dark(Z) :- black(Z),` and prove `black(Q), big(Q).`
3. Now select the fact `black(cat)` and prove `big(cat).`  
   This proof fails!
4. Go back to step 2, and select the second clause of dark, i.e.
   `dark(Z) :- brown(Z),` and prove `brown(Q), big(Q).`
Procedural Meaning of Prolog

brown(bear). big(bear).
gray(elephant). big(elephant).
black(cat). small(cat).
dark(Z) :- black(Z).
dark(Z) :- brown(Z).
dangerous(X) :- dark(X), big(X).

To prove: \textit{? - dangerous(Q)}.

5. Now select \texttt{brown(bear)} and prove \texttt{big(bear)}.
6. Select the fact \texttt{big(bear)}.

There is nothing left to prove, so the proof succeeds
brown(bear).  big(bear).
gray(elephant).  big(elephant).
black(cat).  small(cat).
dark(Z) :- black(Z).
dark(Z) :- brown(Z).
dangerous(X) :- dark(X), big(X).

dangerous(Q) :-
dark(Q), big(Q).
dark(Q) :- black(Q).
dark(Q) :- brown(Q).
black(Q), big(Q)  brown(Q), big(Q)
black(cat).
big(cat)  big(bear)
fail  succeed
The Prolog interpreter works by what is called **backward chaining** (also called *top-down, goal directed*).

- It begins with the thing it is trying to prove and works backwards looking for things that would imply it, until it gets to facts.

- It is also possible to work forward from the facts trying to see if any of the things you can prove from them are what you were looking for.

- This methodology is called **bottom-up resolution**.

- It can be very time-consuming.

- Example: Answer set programming, DLV, Potassco (the Potsdam Answer Set Solving Collection), OntoBroker.

- Fancier logic languages use both kinds of chaining, with special smarts or hints from the user to bound the searches.
Procedural Meaning of Prolog

• When it attempts resolution, the Prolog interpreter pushes the current goal onto a stack, makes the first term in the body the current goal, and goes back to the beginning of the database and starts looking again.

• If it gets through the first goal of a body successfully, the interpreter continues with the next one.

• If it gets all the way through the body, the goal is satisfied and it backs up a level and proceeds.
The Prolog interpreter starts at the beginning of your database (this ordering is part of Prolog, NOT of logic programming in general) and looks for something with which to unify the current goal.

- If it finds a fact, great; it succeeds,
- If it finds a rule, it attempts to satisfy the terms in the body of the rule depth first.
- This process is motivated by the RESOLUTION PRINCIPLE, due to Robinson, 1965:
  - It says that if C1 and C2 are Horn clauses, where C2 represents a true statement and the head of C2 unifies with one of the terms in the body of C1, then we can replace the term in C1 with the body of C2 to obtain another statement that is true if and only if C1 is true.
Procedural Meaning of Prolog

• If it fails to satisfy the terms in the body of a rule, the interpreter **undoes** the unification of the left hand side (BACKTRACKING) (this includes un-instantiating any variables that were given values as a result of the unification) and keeps looking through the database for something else with which to unify.

• If the interpreter gets to the end of database without succeeding, it **backs** out a level (that's how it might **fail** to satisfy something in a body) and continues from there.
Procedural Meaning of Prolog

- **PROLOG IS NOT PURELY DECLARATIVE:**
  - The ordering of the database and the left-to-right pursuit of sub-goals gives a deterministic imperative semantics to searching and backtracking.
  - Changing the order of statements in the database can give you different results:
    - It can lead to infinite loops
    - It can result in inefficiency
Procedural Meaning of Prolog

- Transitive closure with left recursion in Prolog will run into an infinite loop:

  ```prolog
  reach(X,Y) :-
      reach(X,Z),
      edge(Z,Y).
  reach(X,Y) :-
      edge(X,Y).
  ```

  ```prolog
  ?- reach(A,B).
  Infinite loop
  ```
Structures

- If $f$ is an identifier and $t_1, t_2, \ldots, t_n$ are terms, then $f(t_1, t_2, \ldots, t_n)$ is a term.

- In the above, $f$ is called a functor and $t_i$'s are called arguments.

- Structures are used to group related data items together (in some ways similar to struct in C and objects in Java).
  - Structures are used to construct trees (and, as a special case of trees, lists).
Trees

- Example: expression trees:

```
plus (minus (num(3), num(1)), star (num(4), num(2)))
```

- Data structures may have variables AND the same variable may occur multiple times in a data structure.
Matching

- $t_1 = t_2$: finds substitutions for variables in $t_1$ and $t_2$ that make the two terms identical.

(We'll later introduce unification, a related operation that has logical semantics.)

YES, with $X = 1$, $Y = 4$. 

Yes, with $X = 1$, $Y = 4$. 

(c) Paul Fodor (CS Stony Brook) and Elsevier
Matching

• Matching: given two terms, we can ask if they "match" each other
  • A constant matches with itself: \(42\) unifies with \(42\)
  • A variable matches with anything:
    • if it matches with something other than a variable, then it instantiates,
    • if it matches with a variable, then the two variables become associated.
      • \(A=35,\ A=B\) \(\Rightarrow\ B\) becomes \(35\)
      • \(A=B,\ A=35\) \(\Rightarrow\ B\) becomes \(35\)
  • Two structures match if they:
    • Have the same functor,
    • Have the same arity, and
    • Match recursively
      • \(\text{foo}(g(42),37)\) matches with \(\text{foo}(A,37)\),
        \(\text{foo}(g(A),B)\), etc.
Matching

- The general Rules to decide whether two terms $S$ and $T$ *match* are as follows:
  - If $S$ and $T$ are constants, $S = T$ if both are same object
  - If $S$ is a variable and $T$ is anything, $T = S$
  - If $T$ is variable and $S$ is anything, $S = T$
  - If $S$ and $T$ are structures, $S = T$ if
    - $S$ and $T$ have same functor, same arity, and
    - All their corresponding arguments components have to match
Matching

Yes, with $X = 1$, $Y = 4$. 
Matching

No! $X$ cannot be 1 and 4 at the same time.
Matching

• Which of these match?
  • A
  • 100
  • \texttt{func}(B)
  • \texttt{func}(100)
  • \texttt{func}(C, D)
  • \texttt{func}(+(99, 1))
Matching

- Which of these match?
  - A
  - 100
  - func(B)
  - func(100)
  - func(C, D)
  - func(+ (99, 1))
- A matches with 100, func(B), func(100), func(C,D), func(+ (99, 1)).
- 100 matches only with A.
- func(B) matches with A, func(100), func(+ (99, 1)).
- func(C, D) matches with A.
- func(+ (99, 1)) matches with A and func(B).
Accessing arguments of a structure

- Matching is the predominant means for accessing structures arguments
  
  - Let \texttt{date('Sep', 1, 2020)} be a structure used to represent dates, with the month, day and year as the three arguments (\textit{in that order!})

  then \texttt{date(M,D,Y) = date('Sep',1,2020)}. makes

  \[
  M = 'Sep', \quad D = 1, \quad Y = 2020
  \]

  - If we want to get only the day, we can write \texttt{date(_, D, _)} = \texttt{date('Sep', 1, 2020)}. Then we only get: \texttt{D = 1}
Lists

- Prolog uses a special syntax to represent and manipulate lists:
  - \([1,2,3,4]\): represents the list with 1, 2, 3 and 4, respectively.
  - This can also be written as \([1|2,3,4]\): a list with 1 as the **head** (i.e., first element) and \([2,3,4]\) as its **tail** (i.e., the list of remaining elements).
    - If \(X = 1\) and \(Y = [2,3,4]\) then \([X|Y]\) is same as \([1,2,3,4]\).
  - The empty list is represented by \([]\) or **nil**
  - The symbol "|" (**pipe**) and is used to separate the beginning elements of a list from its tail
    - For example: \([1,2|3,4]\) = \([1,2,3|4]\) = \([1,2,3,4]\) = \([1,2,3,4|[]]\) = \([1|2,3,4]\) = \([1|2|3,4]\) = \([1|2|3|4|[]]\)
Lists

• Lists are special cases of trees (syntactic sugar, i.e., internally, they use structures)
  • For instance, the list \([1, 2, 3, 4]\) is represented by the following structure:

![Diagram of a tree representing the list \([1, 2, 3, 4]\)]

  where the function symbol \'. /2 is the list constructor:

  \([1, 2, 3, 4]\) is same as \'. (1, .) (2, .) (3, .) (4, [])

• In XSB:
  \[\text{?- } L = '. (1, [2]).}
  \[L = [1, 2]\]
Lists

• **Strings**: are sequences of characters surrounded by double quotes "abc", "John Smith", "to be, or not to be".

• A string is equivalent to a list of the (numeric) character codes

• **In XSB:**

  \[- \text{?- X} = "abc".\]

  \[X = [97,98,99]\]
Programming with Lists

- `member/2` finds if a given element occurs in a list:
  - The program:
    
    ```prolog
    member(X, [X|_]).
    member(X, [_|Ys]) :-
        member(X,Ys).
    ```
  - Example queries:
    ```prolog
    ?- member(2,[1,2,3]).
    ?- member(X,[l,i,s,t]).
    ?- member(f(X),[f(1),g(2),f(3),h(4)]).
    ?- member(1,L).
    ```
Programming with Lists

- **append/3** concatenate two lists to form the third list:

  - The program:
    - Empty list append L is L:
      \[
      \text{append([], L, L).}
      \]
    - Otherwise, break the first list up into the head X, and the tail L: if L append M is N, then [X|L] append M is [X|N]:
      \[
      \text{append([X|L], M, [X|N]) :- append(L, M, N).}
      \]

  - Example queries:
    - \(?- \text{append([1,2], [3,4], X).}\)
    - \(?- \text{append(X, Y, [1,2,3,4]).}\)
    - \(?- \text{append(X, [3,4], [1,2,3,4]).}\)
    - \(?- \text{append([1,2], Y, [1,2,3,4]).}\)
Is the predicate a function?

**No.** We are not applying arguments to get a result. Instead, we are proving that a theorem holds. Therefore, we can leave any variables unbound.

?- append(L, [2, 3], [1, 2, 3]).
   L = [ 1 ]
?- append([ 1 ], L, [1, 2, 3]).
   L = [2, 3]
?- append(L1, L2, [1, 2, 3]).
   L1 = [], L2 = [1, 2, 3];
   L1 = [1], L2 = [2, 3];
   L1 = [1, 2], L2 = [3];
   L1 = [1, 2, 3], L2 = [];
   no
Append example trace

\[
\text{append}([], L, L).
\text{append}([X|L], M, [X|N]) :- \text{append}(L, M, N).
\]

\[
\text{append}([1,2],[3,4],X)\
\]
Append example trace

append([],L,L).
append([X|L],M,[X|N]) :- append(L,M,N).

append([1,2],[3,4],A)?  
\textbf{X=1, L=[2], M=[3,4], A=[X|N]}
Append example trace

append([],L,L).
append([X|L],M,[X|N]) :- append(L,M,N).

append([2],[3,4],N)?

append([1,2],[3,4],A)?

X=1, L=[2], M=[3,4], A=[X|N]
Append example trace

```prolog
append([], L, L).
append([X|L], M, [X|N']) :- append(L, M, N').
```

Append example trace:

<table>
<thead>
<tr>
<th>Input</th>
<th>Variable Assignments</th>
</tr>
</thead>
<tbody>
<tr>
<td><code>append([2], [3,4], N)</code>?</td>
<td>`X=2, L=[], M=[3,4], N=[2</td>
</tr>
<tr>
<td><code>append([1,2], [3,4], A)</code>?</td>
<td>`X=1, L=[2], M=[3,4], A=[1</td>
</tr>
</tbody>
</table>
Append example trace

\texttt{append([],L,L).}
\texttt{append([X|L],M,[X|N']) :- append(L,M,N').}

\begin{tabular}{|l|}
\hline
\texttt{append([], [3,4], N')?} \\
\hline
\texttt{append([2], [3,4], N)?} & \texttt{X=2, L=[], M=[3,4], N=[2|N']} \\
\hline
\texttt{append([1,2], [3,4], A)?} & \texttt{X=1, L=[2], M=[3,4], A=[1|N]} \\
\hline
\end{tabular}
Append example trace

\[
\text{append}([], L, L).
\]
\[
\text{append}([X|L], M, [X|N']) :- \text{append}(L, M, N').
\]

\[
\begin{array}{ll}
\text{append}([], [3,4], N')? & L = [3,4], N' = L \\
\text{append}([2], [3,4], N)? & X=2, L=[], M=[3,4], N=[2|N'] \\
\text{append}([1,2], [3,4], A)? & X=1, L=[2], M=[3,4], A=[1|N]
\end{array}
\]
Append example trace

\begin{align*}
\text{append}([], L, L). \\
\text{append}([X|L], M, [X|N']) & :- \text{append}(L, M, N').
\end{align*}

<table>
<thead>
<tr>
<th>Step</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>(X=2, L=[], M=[3, 4], N'=[2</td>
</tr>
<tr>
<td>2</td>
<td>(X=1, L=[2], M=[3, 4], A=[1</td>
</tr>
</tbody>
</table>

Answer: \(A=[1, 2, 3, 4]\)
Programming with Lists

- **len/2** to find the length of a list (the first argument):
  - The program:
    ```prolog
    len([], 0).
    len([_|Xs], N+1) :-
      len(Xs, N).
    ```
  - Example queries:
    ```prolog
    ?- len([], X).
      X = 0
    ?- len([l,i,s,t], 4).
      false
    ?- len([l,i,s,t], X).
      X = 0+1+1+1+1
    ```
Arithmetic

?- 1+2 = 3.
false

- In Predicate logic, the basis for Prolog, the only symbols that have a meaning are the predicates themselves.
- In particular, function symbols are uninterpreted: have no special meaning and can only be used to construct data structures.
Arithmetic

- Meaning for arithmetic expressions is given by the built-in predicate "is":
  
  ?- X is 1 + 2.
  succeeds, binding X = 3.

  ?- 3 is 1 + 2.
  succeeds.

- General form: R is E where E is an expression to be evaluated and R is matched with the expression's value.

- Y is X + 1, where X is a free variable, will give an error because X does not (yet) have a value, so, X + 1 cannot be evaluated.
The list length example revisited

- **length/2** finds the length of a list (first argument):
  - The program:
    ```prolog
    length([], 0).
    length([_|Xs], M):-
        length(Xs, N),
        M is N+1.
    ```
  - Example queries:
    ```prolog
    ?- length([], X).
    X = 0
    ?- length([1,i,s,t], 4).
    X = 4
    ?- length([1,i,s,t], X).
    X = 4
    ?- length(List, 4).
    List = [_1, _2, _3, _4]
    ```
Conditional Evaluation

• Conditional operator: the if-then-else construct in Prolog:

  • if A then B else C is written as ( A -> B ; C)

  • To Prolog this means: try A. If you can prove it, go on to prove B and ignore C. If A fails, however, go on to prove C ignoring B.

```prolog
max(X, Y, Z) :-
  ( X =< Y
   -> Z = Y
   ; Z = X
  ).
```

?- max(1,2,X).
X = 2.
Conditional Evaluation

- Consider the computation of \( n! \) (i.e. the factorial of \( n \))

\[
\text{% factorial(+N, -F)}
\]

\[
\text{factorial}(N, F) :- \ldots
\]

- \( N \) is the input parameter and \( F \) is the output parameter!
- The body of the rule specifies how the output is related to the input
  - For factorial, there are two cases: \( N \leq 0 \) and \( N > 0 \)
    - if \( N \leq 0 \), then \( F = 1 \)
    - if \( N > 0 \), then \( F = N \times \text{factorial}(N - 1) \)

\[
\text{factorial}(N, F) :-
\]

\[
(N > 0
\]

\[
\rightarrow \text{N1 is N-1,}
\]

\[
\text{factorial}(N1, F1),
\]

\[
F \text{ is N*F1}
\]

\[
; F = 1
\]

\[
)\]

\[
?- \text{factorial}(12,X).\]

\[
X = 479001600
\]
Imperative features

- Other imperative features: we can think of prolog rules as imperative programs \textit{w/ backtracking}

\begin{verbatim}
program :-
    member(X, [1, 2, 3, 4]),
    write(X),
    nl,
    fail;
    true.

?- program. % prints all solutions
\end{verbatim}

- \texttt{fail}: always fails, causes backtracking
- \texttt{!} is the cut operator: prevents other rules from matching (we will see it later)
Arithmetic Operators

- Integer/Floating Point operators: +, -, *, /
- Automatic detection of Integer/Floating Point
- Integer operators: mod, // (integer division)
- Comparison operators: <, >, =<, >=,
  \( Expr1 =:= Expr2 \) (succeeds if expression \( Expr1 \) evaluates to a number equal to \( Expr2 \)),
  \( Expr1 =\!\!= Expr2 \) (succeeds if expression \( Expr1 \) evaluates to a number non-equal to \( Expr2 \))
Programming with Lists

- Quicksort:

  quicksort([], []).  
  quicksort([X0|Xs], Ys) :-
    partition(X0, Xs, Ls, Gs),
    quicksort(Ls, Ys1),
    quicksort(Gs, Ys2),
    append(Ys1, [X0|Ys2], Ys).

  partition(Pivot, [], [], []).  
  partition(Pivot, [X|Xs], [X|Ys], Zs) :-
    X =< Pivot,
    partition(Pivot, Xs, Ys, Zs).
  partition(Pivot, [X|Xs], Ys, [X|Zs]) :-
    X > Pivot,
    partition(Pivot, Xs, Ys, Zs).
Programming with Lists

• Quicksort:

```prolog
quicksort([], []).
quicksort([X0|Xs], Ys) :-
    partition(X0, Xs, Ls, Gs),
quicksort(Ls, Ys1),
quicksort(Gs, Ys2),
append(Ys1, [X0|Ys2], Ys).

partition(Pivot,([],[],[])).
partition(Pivot,[X|Xs],[X|Ys],Zs) :-
    X =< Pivot,
    !, % cut
    partition(Pivot,Xs,Ys,Zs).
partition(Pivot,[X|Xs],Ys,[X|Zs]) :-
    partition(Pivot,Xs,Ys,Zs).
```
We want to define \texttt{delete/3}, to remove an element from a list:

- \texttt{delete(2, [1,2,3], X)} should succeed with \( X=[1,3] \)
- \texttt{delete(2, [2,1,2], X)} should succeed with \( X=[1,2]; \) \( X = [2,1]; \) fail
- \texttt{delete(X, [1,2,3], [1,3])} should succeed with \( X=2 \)
- \texttt{delete(2, X, [1,3])} should succeed with \( X=[2,1,3]; \) \( X = [1,2,3]; \) \( X=[1,3,2]; \) fail
Algorithm:

- When $X$ is selected from $[X|Ys]$, $Ys$ results as the rest of the list.
- When $X$ is selected from the tail of $[H|Ys]$, $[H|Zs]$ results, where $Zs$ is the result of taking $X$ out of $Ys$. 
Programming with Lists

• The program:

   delete(X, [], _) :- fail. % not needed
   delete(X, [X|Ys], Ys).
   delete(X, [Y|Ys], [Y|Zs]) :-
     delete(X, Ys, Zs).

• Example queries:

   ?- delete(s, [l,i,s,t], Z).
   X = [l, i, t]
   ?- delete(X, [l,i,s,t], Z).
   ?- delete(s, Y, [l,i,t]).
   ?- delete(X, Y, [l,i,s,t]).
Permutations

• Define **permute**/2, to find a permutation of a given list.
  
  • E.g. **permute**([1,2,3], X) should return X=[1,2,3] and upon backtracking, X=[1,3,2], X=[2,1,3], X=[2,3,1], X=[3,1,2], and X=[3,2,1].

  • Hint: What is the relationship between the permutations of [1,2,3] and the permutations of [2,3]?

<table>
<thead>
<tr>
<th>permute([2,3], Y)</th>
<th>permute([1,2,3], Y)</th>
</tr>
</thead>
<tbody>
<tr>
<td>[2,3]</td>
<td>[1,2,3]</td>
</tr>
<tr>
<td></td>
<td>[2,1,3]</td>
</tr>
<tr>
<td></td>
<td>[2,3,1]</td>
</tr>
<tr>
<td>[3,2]</td>
<td>[1,3,2]</td>
</tr>
<tr>
<td></td>
<td>[3,1,2]</td>
</tr>
<tr>
<td></td>
<td>[3,2,1]</td>
</tr>
</tbody>
</table>
Programming with Lists

• The program:

```prolog
permute([], []).
permute([X|Xs], Ys) :-
    permute(Xs, Zs),
    delete(X, Ys, Zs).
```

• Example query:

```prolog
?- permute([1,2,3], X).
```

```
X = [1,2,3];
X = [2,1,3];
X = [2,3,1];
X = [1,3,2] ...
```
The Issue of Efficiency

- Define a predicate, \texttt{rev/2} that finds the reverse of a given list
  - E.g. \texttt{rev([1,2,3],X)} should succeed with \texttt{X=[3,2,1]}
  - Hint: what is the relationship between the reverse of \texttt{[1,2,3]} and the reverse of \texttt{[2,3]}? Answer: \texttt{append([3,2],[1],[3,2,1])}

\begin{verbatim}
rev([], []).
rev([X|Xs], Ys) :- rev(Xs, Zs), append(Zs, [X], Ys).
\end{verbatim}

- How long does it take to evaluate \texttt{rev([1,2,...,n],X)}?
  - \( T(n) = T(n - 1) + \text{time to add 1 element to the end of an n - 1 element list} = T(n - 1) + n - 1 = T(n - 2) + n - 2 + n - 1 = \ldots \)
  - \( T(n) = \Theta(n^2) \) (quadratic)
Making rev/2 faster

- Keep an accumulator: stack all elements seen so far
  - i.e. a list, with elements seen so far in reverse order

- The program:
  
  \[
  \text{rev}(L_1, L_2) :\text{rev}_h(L_1, [], L_2).
  \]

  \[
  \text{rev}_h([X|Xs], \text{AccBefore}, \text{Out}) :\text{rev}_h(Xs, [X|\text{AccBefore}], \text{Out}).
  \]

  \[
  \text{rev}_h([], \text{Acc}, \text{Acc}). \% \text{Base case}
  \]

- Example query:
  
  \[
  ?- \text{rev}([1,2,3], X).
  \]

  will call \text{rev}_h([1,2,3], [], X)

  which calls \text{rev}_h([2,3], [1], X)

  which calls \text{rev}_h([3], [2,1], X)

  which calls \text{rev}_h([], [3,2,1], X)

  which returns \(X = [3,2,1]\)
Tree Traversal

• Assume you have a binary tree, represented by
  • **node/3** facts for internal nodes: \( \text{node}(a, b, c) \) means that \( a \) has \( b \) and \( c \) as children
  • **leaf/1** facts: for leaves: \( \text{leaf}(a) \) means that \( a \) is a leaf

• Example:

\[
\begin{align*}
\text{node}(5, 3, 6). \\
\text{node}(3, 1, 4). \\
\text{leaf}(1). \\
\text{leaf}(4). \\
\text{leaf}(6).
\end{align*}
\]
Tree Traversal

- Write a predicate `preorder/2` that traverses the tree (starting from a given node) and returns the list of nodes in pre-order:

  ```prolog
  preorder(Root, [Root]) :- leaf(Root).
  preorder(Root, [Root|L]) :-
      node(Root, Child1, Child2),
      preorder(Child1, L1),
      preorder(Child2, L2),
      append(L1, L2, L).
  ?- preorder(5, L).
  L = [5, 3, 1, 4, 6]
  ```

- The program takes $O(n^2)$ time to traverse a tree with $n$ nodes. How to append 2 lists in shorter time?
Difference Lists

- The lists in Prolog are singly-linked; hence we can access the first element in constant time, but need to scan the entire list to get the last element.

- However, unlike functional languages like Lisp or SML, we can use variables in data structures:
  - We can exploit this to make lists “open tailed” (also called difference lists in Prolog): end the list with a variable tail and pass that variable, so we can add elements at the end of the list.
Difference Lists

- When \( \mathbf{x} = [1,2,3 | \mathbf{y}] \), \( \mathbf{x} \) is a list with 1, 2, 3 as its first three elements, followed by \( \mathbf{y} \)
- Now if \( \mathbf{y} = [4 | \mathbf{z}] \) then \( \mathbf{x} = [1,2,3,4 | \mathbf{z}] \)
  - We can now think of \( \mathbf{z} \) as “pointing to” the end of \( \mathbf{x} \)
- We can now add an element to the end of \( \mathbf{x} \) in constant time!!
  - And continue adding more elements, e.g. \( \mathbf{z} = [5 | \mathbf{w}] \)
Difference Lists: Conventions

- A *difference list* is represented by two variables: one referring to the entire list, and another to its (uninstantiated) tail
  - e.g. $\mathbf{X} = [1,2,3|\mathbf{Z}]$, $\mathbf{Z}$

- Most Prolog programmers use the notation *List-Tail* to denote a list *List* with tail *Tail*.
  - e.g. $\mathbf{X}-\mathbf{Z}$

- Note that “−” is used as a data structure infix symbol (not used for arithmetic here)
Difference Lists

- Append 2 open ended lists:

\[ \text{dappend}(X, T, Y, T_2, L, T_3) :- \\
  T = Y, \\
  T_2 = T_3, \\
  L = X. \]

?- dappend([1,2,3|T],T, [4,5,6|T2],T2, L,T3).
L = [1,2,3,4,5,6|T3]

- Simplified version:

\[ \text{dappend}(X, T, T, T_2, X, T_2). \]

- More simplified notation (with "-"):

\[ \text{dappend}(X-T, T-T_2, X-T_2). \]

?- dappend([1,2,3|T]-T, [4,5,6|T2]-T2, L-T3).
L = [1,2,3,4,5,6|T2]
Difference Lists

• Add an element at the end of a list:

\[
\text{add}(L-T, X, L2-T2) :-
T = [X|T2],
L = L2.
\]

?- add([1,2,3|T]-T, 4, L-T2).
L = [1,2,3,4|T2]

• We can simplify it as:

\[
\text{add}(L-T, X, L-T2) :-
T = [X|T2].
\]

• This can be simplified more like:

\[
\text{add}(L-[X|T2], X, L-T2).
\]

• Alternative using \text{dappend}:

\[
\text{add}(L-T, X, L-T2) :-
dappend(L-T, [X|T2]-T2, L-T2).
\]
Difference Lists

• Check if a list is a palindrome:

```prolog
palindrome(X) :-
    palindromeHelp(X-[]).
palindromeHelp(A-A). % an empty list
palindromeHelp([_|A]-A). % 1-element list
palindromeHelp([C|A]-D) :-
    palindromeHelp(A-B),
    B=[C|D].
?- palindrome([1,2,2,1]).
yes
?- palindrome([1,2,3,2,1]).
yes
?- palindrome([1,2,3,4,5]).
no
```
preorder1(Node, List, Tail) :-
    node(Node, Child1, Child2),
    List = [Node|List1],
    preorder1(Child1, List1, Tail1),
    preorder1(Child2, Tail1, Tail).
preorder1(Node, [Node|Tail], Tail) :-
    leaf(Node).
preorder(Node, List) :-
    preorder1(Node, List, []).

• The program takes $O(n)$ time to traverse a tree with $n$ nodes
The preorder traversal program may be rewritten as:

```prolog
preorder1(Node, [Node|L]-T) :-
  node(Node, Child1, Child2),
  preorder1(Child1, L-T1),
  preorder1(Child2, T1-T).
preorder1(Node, [Node|T]-T).
```
Difference Lists: Conventions

- The inorder traversal program:

\[
\text{inorder1}(	ext{Node}, \text{L-T}) : - \\
\text{node}(\text{Node}, \text{Child1}, \text{Child2}), \\
\text{inorder1}(	ext{Child1}, \text{L-T1}), \\
\text{T1} = \text{[Node|T2]}, \\
\text{inorder1}(	ext{Child2}, \text{T2-T}). \\
\text{inorder1}(	ext{Node}, \text{[Node|T]-T}). \\
\text{inorder}(	ext{Node,L}) : - \\
\text{inorder1}(	ext{Node}, \text{L-[]}).
\]
Difference Lists: Conventions

- The postorder traversal program:

\[
\text{postorder1}(\text{Node}, \text{L-T}) :\neg \\
\text{node}(\text{Node}, \text{Child1}, \text{Child2}), \\
\text{postorder1}(\text{Child1}, \text{L-T1}), \\
\text{postorder1}(\text{Child2}, \text{T1-T2}), \\
\text{T2} = [\text{Node}|\text{T}]. \\
\text{postorder1}(\text{Node}, [\text{Node}|\text{T}]\text{-T}). \\
\text{postorder}(\text{Node}, \text{L}) :\neg \\
\text{postorder1}(\text{Node}, \text{L-[]}).
\]
There are several ways to represent graphs in Prolog:

- represent each edge separately as one clause (fact):
  
  ```prolog
  edge(a,b).
  edge(b,c).
  ```

- isolated nodes cannot be represented, unless we have also `node/1` facts

- the whole graph as one data object: as a pair of two sets (nodes and edges): `graph([a,b,c,d,f,g], [e(a,b), e(b,c), e(b,f)])`

- list of arcs: `[a-b, b-c, b-f]`

- adjacency-list: `[n(a, [b]), n(b, [c,f]), n(d, [])]`
Graphs in Prolog

Path from one node to another one:

- A predicate `path(+G,+A,+B,−P)` to find an acyclic path `P` from node `A` to node `B` in the graph `G`.
- The predicate should return all paths via backtracking.
- We will solve it using the graph as a data object, like in `graph([a,b,c,d,f,g], [e(a,b), e(b,c), e(b,f)])`.
Graphs in Prolog

• \textbf{adjacent} for \textit{directed edges}:

\begin{verbatim}
adjacent(X,Y,graph(_,Es)) :-
    member(e(X,Y),Es).
\end{verbatim}

• \textbf{adjacent} for \textit{undirected edges} (ie. no distinction between the two vertices associated with each edge):

\begin{verbatim}
adjacent(X,Y,graph(_,Es)) :-
    member(e(X,Y),Es).
adjacent(X,Y,graph(_,Es)) :-
    member(e(Y,X),Es).
\end{verbatim}
• Path from one node to another one:

```
path(G,A,B,P) :-
    pathHelper(G,A,[B],P).
% Base case
pathHelper(_,A,[A|P1],[A|P1]).
pathHelper(G,A,[Y|P1],P) :-
    adjacent(X,Y,G),
    \+ member(X,[Y|P1]),
    pathHelper(G,A,[X,Y|P1],P).
```
Graphs in Prolog

• Cycle from a given node in a directed graph:
  • a predicate `cycle(G,A,Cycle)` to find a closed path (cycle) `Cycle` starting at a given node `A` in the graph `G`
  • The predicate should return all cycles via backtracking

\[
\text{cycle}(G,A,Cycle) :- \\
\text{adjacent}(A,B,G), \\
\text{path}(G,B,A,P1), \\
\text{Cycle} = [A|P1].
\]
Complete program in XSB

:- import member/2 from basics.

adjacent(X,Y,graph(_,Es)) :-
    member(e(X,Y),Es).

path(G,A,B,P) :-
    pathHelper(G,A,[B],P).

pathHelper(_,A,[A|P1],[A|P1]).

pathHelper(G,A,[Y|P1],P) :-
    adjacent(X,Y,G),
    member(X,[Y|P1]),
    pathHelper(G,A,[X,Y|P1],P).

cycle(G,A,Cycle) :-
    adjacent(A,B,G),
    path(G,B,A,P),
    Cycle = [A|P].

?- Graph = graph([a,b,c,d,f,g],
                 [e(a,b), e(b,c), e(c,a), e(a,e), e(e,a)]),
    cycle(Graph,a,Cycle),
    writeln(Cycle),
    fail; true.
Aggregates in XSB

- **setof(Template,Goal,Set)**: Set is the set of all instances of Template such that Goal is provable.

- **findall(Template,Goal,List)** is similar to predicate `bagof/3`, except that variables in Goal that do not occur in Template are treated as existential, and alternative lists are not returned for different bindings of such variables.

- **bagof(Template,Goal,Bag)** has the same semantics as `setof/3` except that the third argument returns an unsorted list that may contain duplicates. X^Goal will not bind X.

- **tfindall(Template,Goal,List)** is similar to predicate `findall/3`, but the Goal must be a call to a single tabled predicate.
Aggregates in XSB

\[ p(1,1). \]
\[ p(1,2). \]
\[ p(2,1). \]
\[ \text{?- setof}(Y, p(X,Y), L). \]
\[ L=[1,2] \]
\[ \text{?- findall}(Y, p(X,Y), L). \]
\[ L=[1,2,1] \]
\[ \text{?- bagof}(Y, p(X,Y), L). \]
\[ X=1, \ L=[1,2] \ ; \]
\[ X=2, \ L=[1] \ ; \]
\[ \text{fail} \]
XSB Prolog

• Negation: \+ is negation-as-failure
• Another negation called \texttt{tnot} (\texttt{TABLING = memoization})
  • Use: \ldots: \texttt{tnot(foobar(X))}.
  • All variables under the scope of \texttt{tnot} must also occur to the left of that scope in the body of the rule in other positive relations:
    • Ok: \ldots: \texttt{\texttt{-p(X,Y),tnot(foobar(X,Y))},...}
    • Not ok: \ldots: \texttt{\texttt{-p(X,Z),tnot(foobar(X,Y))},...}
• XSB also supports Datalog:
  \texttt{:- auto_table.}
  at the top of the program file
XSB Prolog

- Read/write from and to files:
  - Edinburgh style:

```prolog
?- tell('a.txt'), write('Hello, World!'), told.

?- see('a.txt'), read(X), seen.
```
XSB Prolog

- Read/write from and to files:
- ISO style:

```prolog
?- open('a.txt', write, X),
   write(X,'Hello, World!'),
   close(X).
```
Cut (logic programming)

- Cut (! in Prolog) is a goal which always succeeds, but cannot be backtracked past:
  \[
  \text{max}(X,Y,Y) : - X \leq Y, !. \\
  \text{max}(X, _, X).
  \]
- cut says “stop looking for alternatives”
- no check is needed in the second rule anymore because if we got there, then \( X \leq Y \) must have failed, so \( X > Y \) must be true.
- Red cut: if someone deletes !, then the rule is incorrect - above
- Green cut: if someone deletes !, then the rule is correct
  \[
  \text{max}(X,Y,Y) : - X \leq Y, !. \\
  \text{max}(X,Y,X) : - X > Y.
  \]
  - by explicitly writing \( X > Y \), it guarantees that the second rule will always work even if the first one is removed by accident or changed (cut is deleted)
Cut (logic programming)

- No backtracking pass the guard, but ok after:

```
p(a). p(b).
q(a). q(b). q(c).

?- p(X),!.
X=a ;
no

?- p(X),!,q(Y).
X=a, Y=a ;
X=a, Y=b ;
X=a, Y=c ;
no
```
Testing types

- **atom(X)**
  Tests whether X is bound to a symbolic atom
  
  ```
  ?- atom(a).
  yes
  ?- atom(3).
  no
  ```

- **integer(X)**
  Tests whether X is bound to an integer

- **real(X)**
  Tests whether X is bound to a real number
Testing for variables

- **is_list**(L)
  Tests whether **L** is bound to a list

- **ground**(G)
  Tests whether **G** has unbound logical variables

- **var**(X)
  Tests whether **X** is bound to a Prolog variable
Control / Meta-predicates

- **call(P)**
  Force P to be a goal; succeed if P does, else fail

- **clause(H,B)**
  Retrieves clauses from memory whose head matches H and body matches B. H must be sufficiently instantiated to determine the main predicate of the head

- **copy_term(P,NewP)**
  Creates a new copy of the first parameter (with new variables)
  - It is used in iteration through non-ground clauses, so that the original calls are not bound to values
Control / Meta-predicates

• Write a Prolog relation

\[ \text{map(} \text{BinaryRelation, InputList, OutputList)} \]

which applies a binary relation on each of the elements of the list \text{InputList} as the first argument and collects the second argument in the result list.

• Example:

\[ \text{?- map(} \text{inc1(X,Y)}, [5,6], R) \]. \ \text{returns } R=[6,7] \]

where \text{inc1(X,Y)} was defined as:

\[ \text{inc1(X,Y)} : - \]

\[ \text{Y is X+1.} \]
Control / Meta-predicates

\[
\text{map}(_{-}\text{BinaryCall}, [], []). \\
\text{map}(\text{BinaryCall}, [X|T], [Y|T2]) :- \\
\quad \text{copy_term(}\text{BinaryCall}, \text{BinaryCall2}), \\
\quad \text{BinaryCall2} =_{..} [_{F}, X, Y], \\
\quad \text{call(}\text{BinaryCall2}), \\
\quad \text{map}(\text{BinaryCall}, T, T2). \\
\text{inc1}(X, Y) :- \\
\quad Y \text{ is } X + 1. \\
?\text{- map(}\text{inc1}(X, Y), [5, 6], R). \\
\text{R} = [6, 7]
\]
Control / Meta-predicates

\[ \text{square}(X,Y) :- \]
\[ \text{Y is } X*X. \]

?- \text{map}(\text{square}(E, E2), [2, 3, 1], R).
\[ \text{R} = [4, 9, 1]; \]
\[ \text{no} \]
Use the relation \texttt{map} to implement a relation \texttt{pairAll(E,L,L2)} which pairs the element \texttt{E} with each element of the list \texttt{L} to obtain \texttt{L2}.

Examples:

\texttt{?- pairAll(1,[2,3,1], L2).}
returns \texttt{L2=[[1,2], [1,3], [1,1]]}

\texttt{?- pairAll(1,[], L2).}
returns \texttt{L2=[]}. 
Control / Meta-predicates

\[
\text{pair}(E2, \ (_E1,E2)).
\]

\[
\text{pairAll}(E,L,L2):- \\
\quad \text{map}(\text{pair}(E2, \ (E,E2)), \ L, \ L2).
\]

?- \text{pairAll}(1, \ [2,3,1], \ R).
R = [(1,2),(1,3),(1,1)]
Assert and retract

- **asserta (C)**
  Assert clause \( C \) into database above other clauses with the same predicate.

- **assertz (C), assert (C)**
  Assert clause \( C \) into database below other clauses with the same predicate.

- **retract (C)**
  Retract \( C \) from the database. \( C \) must be sufficiently instantiated to determine the predicate.
Prolog terms

• `functor (E, F, N)`
  E must be bound to a functor expression of the form '
  f(...}'. F will be bound to 'f', and N will be bound to the number of arguments that f has.

• `arg (N, E, A)`
  E must be bound to a functor expression, N is a whole number, and A will be bound to the Nth argument of E
Prolog terms and clauses

- \(=..\)
  
  converts between term and list. For example,

?\(-\) parent(a,X) =.. L.
L = [parent, a, _X001]

?- [1] =.. X.
X = [.,1,[]]